http://www.mapleprimes.com/questions/151602-Solve-The-Following-System-Of-Equations
due to Kitonum

```
> restart; interface(version);
  sys:=[sqrt(sin(x)^2+1/sin(x)^2)+sqrt(cos(y)^2+1/cos(y)^2) = sqrt(20*y/(x+y)),
  sqrt(sin(y)^2+1/sin(y)^2)+sqrt(cos(x)^2+1/cos(x)^2) = sqrt(20*x/(x+y))];
```

Classic Worksheet Interface, Maple 17.00, Windows, Feb 21 2013, Build ID 813473 **(1)**

$$sys := \left[\sqrt{\sin(x)^2+\frac{1}{\sin(x)^2}}+\sqrt{\cos(y)^2+\frac{1}{\cos(y)^2}} = 2\sqrt{5}\sqrt{\frac{y}{x+y}}, \sqrt{\sin(y)^2+\frac{1}{\sin(y)^2}}+\sqrt{\cos(x)^2+\frac{1}{\cos(x)^2}} = 2\sqrt{5}\sqrt{\frac{x}{x+y}}\right]$$

The LHS is defined except pols. For the RHS: if 0 < x+y then 0 <= y and 0 <= x to be non-complex from the 1st resp. 2nd equation, similar for the negative case. Since x = 0 = y is forbidden by the LHS one has: both x,y are either positive or negative.

And since the system is symmetric w.r.t. signs one can use

```
> sin(x)^2 = p: %, isolate(%, x), 0 < p, p<=1; #x in {solve(%%,x)};
```

$$\sin(x)^2 = p, x = \arcsin\left(\sqrt{p}\right), 0 < p, p \le 1 \quad \textbf{(2)}$$

```
> 'sys[1]+sys[2]':
  'eval(%, x = arcsin(p^(1/2)))':
  'eval(%, y = arcsin(q^(1/2)))';
  SYS:=%;
  LHS, RHS:=lhs(SYS), rhs(SYS);
```

$$\left(sys_1+sys_2\right)\Big|_{x=\arcsin\left(\sqrt{p}\right)}\Big|_{y=\arcsin\left(\sqrt{q}\right)} \quad \textbf{(3)}$$

$$SYS := \sqrt{p+\frac{1}{p}}+\sqrt{1-q+\frac{1}{1-q}}+\sqrt{q+\frac{1}{q}}+\sqrt{1-p+\frac{1}{1-p}} = 2\sqrt{5}\sqrt{\frac{\arcsin\left(\sqrt{q}\right)}{\arcsin\left(\sqrt{p}\right)+\arcsin\left(\sqrt{q}\right)}}+2\sqrt{5}\sqrt{\frac{\arcsin\left(\sqrt{p}\right)}{\arcsin\left(\sqrt{p}\right)+\arcsin\left(\sqrt{q}\right)}}$$

$$LHS, RHS := \sqrt{p+\frac{1}{p}}+\sqrt{1-q+\frac{1}{1-q}}+\sqrt{q+\frac{1}{q}}+\sqrt{1-p+\frac{1}{1-p}}, 2\sqrt{5}\sqrt{\frac{\arcsin\left(\sqrt{q}\right)}{\arcsin\left(\sqrt{p}\right)+\arcsin\left(\sqrt{q}\right)}}+2\sqrt{5}\sqrt{\frac{\arcsin\left(\sqrt{p}\right)}{\arcsin\left(\sqrt{p}\right)+\arcsin\left(\sqrt{q}\right)}}$$

Plotting shows: the LHS is increasing, the RHS is decreasing

```
> 'LHS':
  plot3d(min(%, 10), p=0..1,q=0..1, axes=boxed, orientation=[40,80], color=
  "LightBlue", title = "blue=LHS, red = RHS"): pa:=%:
  'RHS':
  plot3d(min(%, 10), p=0..1,q=0..1, axes=boxed, orientation=[20,60], color=
  "LightSalmon"): pb:=%:
```

```
plots[display](pa,pb);
```

blue=LHS, red = RHS




```
> [minimize(LHS, p = 0 .. 1, q= 0 .. 1, location)]: %[2]; #evalf(%);
```

$$\left\{\left[\left\{p=\frac{1}{2}, q=\frac{1}{2}\right\}, 2\sqrt{5}\sqrt{2}\right]\right\} \tag{4}$$

The minimum is achieved for p = 1/2 = q, which means to solve $\sin(t)^2 = \frac{1}{2}$. Which is the set $\frac{1}{4}\pi + \frac{1}{2}\pi n$, n integer:

```
> 'solve(sin(t)^2=1/2, x)';
  #solve(sin(t)^2=1/2, t, allsolutions);
  solve([sin(t)^2=1/2, -2*Pi <= t, t <= 2*Pi], AllSolutions, explicit): sort([%])
  : `...`, op(%), `...`;
  #plot(sin(t)^2-1/2, tickmarks=[piticks, decimalticks]);
```

$$solve\left(\sin(t)^2 = \frac{1}{2}, x\right)$$

$$\ldots, \left\{t=-\frac{7}{4}\pi\right\}, \left\{t=-\frac{5}{4}\pi\right\}, \left\{t=-\frac{3}{4}\pi\right\}, \left\{t=-\frac{1}{4}\pi\right\}, \left\{t=\frac{1}{4}\pi\right\}, \left\{t=\frac{3}{4}\pi\right\}, \left\{t=\frac{5}{4}\pi\right\}, \left\{t=\frac{7}{4}\pi\right\}, \ldots \tag{5}$$

These are solutions for the system:

```
> 'subs(x = Pi/4 + Pi/2*n, y = Pi/4 + Pi/2*n, sys)';
  combine(%) assuming n::integer;
```

```
#simplify(%) assuming n::integer;
```

$$subs\left(x = \frac{1}{4}\pi + \frac{1}{2}\pi n, y = \frac{1}{4}\pi + \frac{1}{2}\pi n, sys\right) \quad \mathbf{(6)}$$

$$\left[\sqrt{10} = \sqrt{10}, \sqrt{10} = \sqrt{10}\right]$$

It remains to show: these are all solutions. And for that it is enough to show, that the maximum for the RHS is unique and equals the minimum of the LHS.

```
> eval(rhs(sys[1]+sys[2]), y = u*x); # 0 < u, due to same signs
  [maximize(%, u = 0 .. infinity, location)]: %[2];
```

$$2\sqrt{5}\sqrt{\frac{u\,x}{u\,x + x}} + 2\sqrt{5}\sqrt{\frac{x}{u\,x + x}} \quad \mathbf{(7)}$$

$$\left\{\left[\{u = 1\}, 2\sqrt{5}\sqrt{2}\right]\right\}$$

Which means: we must have x=y and the (!) maximum RHS equals the minimum LHS.